
Graphical Presentation of Survival Analysis

Joanne C. Zhou, GlaxoSmithKline Pharmaceuticals,
Collegeville, PA

ABSTRACT

Survival analysis and graphical
presentation of the survival analysis
result are not uncommon in
pharmaceutical industry.  In this paper
we use simulated survival data to
demonstrate the graphical presentation
of Kaplan-Meier survival analysis
result by using Proc Lifetest, Proc
Gplot and Proc Greplay.

SURVIVAL ANALYSIS AND PRESENTATION

Simulated Survival Data

Following is the code we used to
generate the survival data.  In these
data, we assume there are 2 treatment
groups, namely study drug and placebo.
And there are 100 patients in each
treatment group. The survival time is
exponentially distributed and the
treatment effect is 20 units for study
drug and 10 units for placebo.  The
censoring indicator variable CENSOR is
coded 1 for the patients who were
censored and 0 for the patients who had
event.

options nocenter;

proc format;
  value trt
    1 = 'Study Drug'
    2 = 'Placebo';
run;

data surv;
  do trt = 1 to 2;
    do pat = 1 to 100;
      if trt = 1 then do;
        surv = -log(ranuni(0))*20;
        if mod(pat, 5) then censor = 0;
        else censor = 1;
      end;
      else do;

        if mod(pat, 7) then censor = 0;
        else censor = 1;
        surv = -log(ranuni(0))*10;
      end;
      drop pat;
      patient = put(pat, z5.);
      output;
    end;
  end;
run;

Sample data:

patient    trt      surv     censor

 00001      1     75.7023       0
 00002      1      7.0416       0
 00003      1      5.1007       0
 00004      1      5.8182       0
 00005      1      3.8551       1
 00006      1      6.5943       0
 00007      1     11.2427       0
 00008      1     38.2103       0
 00009      1     44.6357       0
 00010      1     23.2238       1
.
.
.

Survival Analysis

The Kaplan-Meier survival estimates
were created by calling Proc Lifetest
procedure.

ods listing close;
ods output  ProductLimitEstimates  = s_est
            CensoredSummary = cen_sum
            HomTests = homtest
            Quartiles = quart;
  proc lifetest data = surv;
    time surv*censor(1);
    strata trt;
    id patient;
    format trt trt.;
  run;

ods output close;
ods listing;

Extraction of Statistics

SAS ODS provides great convenience for
extraction of Log-Rank p-value, median
survival time and other statistics.
Here we convert these statistics into
macro variables so that we can annotate
them onto the survival curves later on.

/**************************************
   Populate the censored patients
   with appropriate KM estimates.
**************************************/

data s_est;
  set s_est;
  retain pr_s;
  if censor = 0 then pr_s = survival;
  surviva = .;
  if censor = 1 then do;
    surviva = pr_s;
    survival = .;
  end;
run;

/**************************************
   Convert the total number of
   patients, number of patient with
   event, number of patients with
   censoring into macro variables.
**************************************/

proc sort data = cen_sum;
  by trt;
  where control_var is missing;
run;

data _null_;
  set cen_sum end = last;
  pctevt = 100 - pctcens;
  call symput('groupf'||compress(_n_),
               compress(put(trt, trt.)));
  call symput('ng'||compress(_n_),
               compress(put(total, 8.)));
  call symput('ngevt'||compress(_n_),
               compress(put(failed, 8.)));
  call symput('pgevt'||compress(_n_),
               compress(put(pctevt, 8.2)));
  call symput('ngcen'||compress(_n_),
               compress(put(censored, 8.)));
  call symput('pgcen'||compress(_n_),
               compress(put(pctcens, 8.2)));

run;

/**************************************
   Convert Log-Rank test p-value
   into macro variable.
**************************************/

proc sort data = homtest;
  where test = 'Log-Rank';
  by test;
run;

data _null_;
  set homtest;
  call symput('logrp', compress(put(probchisq,
               pvalue6.4)));
run;

/**************************************
   Convert median survival estimates
   and its 95% CI into variables.
**************************************/

proc sort data = quart;
  by trt;
  where percent = 50;
run;

data _null_;
  set quart;
  call symput('gmed'||compress(_n_),
               compress(put(estimate, 8.2)));
  call symput('llmit'||compress(_n_),
               compress(put(lowerlimit, 8.2)));
  call symput('ulmit'||compress(_n_),
               compress(put(upperlimit, 8.2)));
run;

Plot the Kaplan-Meier Estimates

Proc Gplot is used to plot the survival
estimates against the observed survival
time.  Notice that we call the Proc
Gplot twice.  One is for the event
curve and the other for the censored
curve.  The survival statistics
including log-rank p-value are put on
the graph by note statements. In
addition, the name statement is used to
create two different graphs in order to
replay the two graphs onto one graph
which will have both event and censored
curves.

SUGI 27 Posters



3

/********
   Gplot
********/

title1 h = 3 "Kaplan-Meier Survival Estimate";

filename event "c:/survival/lifetest_event.cgm";
filename censor "c:/survival/lifetest_censo.cgm";

axis1  c = black value = (h = 2.2) minor = none
       order = (0 to 1 by 0.1) width = 2
       label=(angle=90 h = 2.4 'Kaplan-Meier
Estimate');

axis2  c = black value = (h = 2.2)
       order = (0 to 80 by 10) width = 2
         minor = none label= ( h = 2.4 "Survival
Time (Months)");

legend value = (height = 1.8) cborder  = black
         label    = none
         position = (TOP inside RIGHT)
         mode     = share;

goptions reset = symbol rotate = landscape
         device  = cgmmwwc gsfname = event
         ctext = black cback = white
         gsfmode = replace ftext=swissb
         gsflen=80 display autofeed
         gunit   = pct;

proc gplot data = s_est gout = work.mygraf;
  note height = 1.3 j = left font = swissbu
  move = (62,  74) "&groupf1"
  move = (48,  71) "No of Patients"
  move = (48,  69) "Death"
  move = (48,  67) "Censored"
  move = (48,  65) "Median Survival"
  move = (48,  63) "95% CI"
  move = (48,  60) "Log Rank P-Value : &logrp"
  move = (64,  71) "&ng1"
  move = (64,  69) "&ngevt1 (&pgevt1.%)"
  move = (64,  67) "&ngcen1 (&pgcen1.%)"
  move = (64,  65) "&gmed1 Mos."
  move = (64,  63)  "[&llmit1, &ulmit1]"
  move = (76, 74)  "&groupf2"
  move = (78, 71)  "&ng2"
  move = (78, 69)  "&ngevt2 (&pgevt2..%)"
  move = (78, 67)  "&ngcen2 (&pgcen2..%)"
  move = (78, 65)  "&gmed2 Mos."
  move = (78, 63)  "[&llmit2, &ulmit2]" ;

symbol1 V = none I = steplj L = 1 c = black
        H = 1.5 W = 3.5;
symbol2 V = none I = steplj L = 2 c = black
        H = 1.5 W = 3.5;

    plot survival*surv = trt
           / name = 'g1'
             haxis=axis2
             vaxis=axis1
             legend = legend ;
         format survival 8.1 surv 8.;
run;

goptions reset = symbol gsfname = censor
         gsfmode = replace;

proc gplot data = s_est gout = work.mygraf;
     symbol1 V = dot I = none c = black
             H = 1.5 W = 3.5;
     symbol2 V = circle I = none c = black
             H = 1.5 W = 3.5;
      plot surviva*surv = trt
           / name = 'g2'
             haxis=axis2
             vaxis=axis1
             legend = legend;
            format surviva 8.1 surv 8.;
run;

Overlay the Survival Curves

In order to show both the event
patients and the censored patients on
the survival curves, the Proc Greplay
is used to overlay the two survival
curves.

filename wordfile "c:/survival/lifetest.cgm";

goptions reset = symbol gsfname = wordfile
         gsfmode = replace;

proc greplay  igout = work.mygraf
                  tc  = work.newtwo nofs;
  tdef newtwo  des = 'Survival Curves'
     1/llx =  0    lly = 0
       ulx =  0    uly = 100
       urx = 100   ury = 100
       lrx = 100   lry = 0;
       template newtwo;
       treplay
         1:g1
         1:g2;
     quit;
run;

proc catalog cat = work.mygraf;
  delete g1 g2/entrytype = grseg;
run;

CONCLUSIONS

Gplot, Greplay and annotation
techniques are powerful tools in
presenting the survival analysis
results graphically.  We can not only
show the survival curves, but also the
survival statistics on one graph.  In
addition, with different devices and
colors specified in Goptions, we can
convert the regular survival analysis
results into colorful slides for
example.

